

最適化シミュレーション演習 第7回

1. Benders 分解法で使用するモデルファイル

使用するモデルファイル fl-master.mod では2つの問題を含んでいることに注意せよ。

```
#problem MASTER; #マスター問題
param NPLANT ;
param NDEMAND ;
param alpha {k in 1..NCUT, i in 1..NPLANT};    #最適性カットの x 係数
param beta {k in 1..NCUT};                      #最適性カットの定数部
param alpha1 {k in 1..NfeaCUT, i in 1..NPLANT}; #省略
param beta1 {k in 1..NfeaCUT};                  #省略
param xx {i in 1..NPLANT};                      #マスター問題で得られた x の値
param fixcost {i in 1..NPLANT}>=0;
param demand {j in 1..NDEMAND}>=0;
param caplimit {i in 1..NPLANT}>=0;
param l {i in 1..NPLANT};
param u {i in 1..NPLANT};
var x {i in 1..NPLANT} >=0, integer;            #変数 x および  $\theta$ 
var theta >=0;

minimize costmaster:    #MASTER 問題では  $\sum fx + \theta$  を最小化
sum {i in 1..NPLANT} fixcost[i]*x[i] +theta;

subject to luz {i in 1..NPLANT}:                #変数 x の上下限制約
    l[i] <= x[i] <= u[i];

subject to feascut {k in 1..NfeaCUT}:            #省略：実行可能性カット
0>=sum{i in 1..NPLANT} alpha1[k,i]*x[i] +beta1[k];

subject to optcut {k in 1..NCUT}:                #最適性カット、最初は NCUT=0 なので制約の本数=0
theta >= sum{i in 1..NPLANT} alpha[k,i]*x[i] +beta[k];

subject to feasibility: #Q(x)を定義する LP が必ず実行可能になるように追加した制約
    sum {i in 1..NPLANT} caplimit[i]*x[i] >= sum {j in 1..NDEMAND} demand[j];

#problem SECOND;    #Q(x)を求める問題
var y {i in 1..NPLANT, j in 1..NDEMAND}>=0;    #輸送量 y が変数
param varcost {i in 1..NPLANT, j in 1..NDEMAND}>=0;

minimize secondcost:    #輸送費用の最小化
sum {i in 1..NPLANT, j in 1..NDEMAND} varcost[i,j]*y[i,j];

subject to capacity {i in 1..NPLANT}:            #輸送量 ≤ 容量となる制約 (MASTER 問題の解は xx)
    -sum {j in 1..NDEMAND} y[i,j] >= -caplimit[i]*xx[i];

subject to dem1 {j in 1..NDEMAND}:                #需要を満足させる制約
    sum {i in 1..NPLANT} y[i,j] = demand[j];
```

2. Benders 分解法のアルゴリズム

```
option presolve 0; #前処理を行わない

param NCUT;      #最適性カット本数=0
param NfeaCUT;   #省略：実行可能性カット本数=0
let NCUT:=0;
let NfeaCUT:=0;
model fl-master.mod ;
data fl.dat ;
option solver cplexamp;
problem MASTER: x, theta, costmaster, luz, feascut, optcut, feasibility; #前頁 MASTER
problem SECOND: y, secondcost, capacity, dem1;                        #前頁 Q(x)
problem DIRECTION: mu, pi, directgen, direct_mupi, direct_mu, direct_pi; #省略

param UB; #暫定目的関数値
param LB; #下界値
let UB:=1e12;
let LB:=0;

display UB, LB;

repeat
{
    printf: "Yn##### NCUT=%d Yn", NCUT;
    printf: "Yn##### NCUT=%d Yn", NCUT > fl-master.sol;
    solve MASTER;          #MASTER 問題の求解
    for {i in 1..NPLANT}
    {
        let xx[i]:=x[i];   #MASTER 問題の解を xx
    }

    if (costmaster>LB) then    #下界値の更新
    { let LB:=costmaster;}

    solve SECOND;

    if ( solve_result="infeasible") then    #以下本ページ内は省略：実行可能性カット
    {
        solve DIRECTION;

        let NfeaCUT:=NfeaCUT+1;
        for {i in 1..NPLANT}
        {
            let alpha1[NfeaCUT, i]:=caplimit[i]*mu[i];
        }
        let beta1[NfeaCUT]:=sum{j in 1..NDEMAND} demand[j]*pi[j];
    }
}
```

```

else if ( solve_result="solved" and theta < secondcost) then #最適性カット生成
{
    let NCUT:=NCUT+1;                                #カット本数追加
    for {i in 1..NPLANT}
    {
        let alpha[NCUT,i]:=-caplimit[i]*capacity[i].dual;    #係数  $\mu$  定義
    }
    let beta[NCUT]:=sum{j in 1..NDEMAND} (demand[j]*dem1[j].dual); #係数  $\pi$  定義

    if (sum {i in 1..NPLANT} fixcost[i]*x[i]+secondcost < UB) then #上界の更新
    { let UB:=sum {i in 1..NPLANT} fixcost[i]*x[i]+secondcost;}
    }
    display x, theta > fl-master.sol;
    display UB, LB;
    if (LB > (1-1e-2)*UB) then #終了判定条件
    {
        break;
    }
}

printf: "¥n##### FINISHED ¥n";
printf: "¥n##### FINIFHED ¥n" > fl-master.sol;

solve MASTER >fl-master.sol;
expand > fl-master.sol;
display costmaster > fl-master.sol;
display x > fl-master.sol;
display y > fl-master.sol;

display _ampl_time >fl-master.sol;
display _total_solve_time >fl-master.sol;
quit;

```

課題：施設配置候補地数や需要地数の異なる問題を解き、(Benders 分解を用いない) 通常の方法と比較せよ。特に、データの与え方（施設の容量と固定費を大きく）を変化すると、どのような結果が得られるか考察せよ。

最終課題：

本演習において取り扱った適当な問題を取りあげるか、または異なる問題をきちんと提示せよ。その上、数理計画モデルとして定式化し、適切なデータを作成（収集）せよ（データの与え方は、これまでに配布した dataconst.run などのプログラムを参考にすること）。本演習で扱った分枝限定法などの解法を用いて、結果を得て、それを評価せよ。